```
model {

   sw[1]<- 0
   for(i in 1:N) {
      p[i,1] <- 1
      for (j in 1:nc[i]-1) {
         r[i,j] ~ dbin(q[i,j],n[i,j])
         q[i,j] <- 1-(p[i,C[i,j+1]]/p[i,C[i,j]])
         z.index[i,j]<- C[i,j+1]-1
         theta[i,j] <- mu[s[i]] + delta[i]*(1-equals(t[i],b[i])) + z[z.index[i,j]] +
                        betaplac*(mu[s[i]]-mu_m)*(1-equals(t[i],1)) +
                        (beta[t[i]]-beta[t[1]]) * (1-equals(t[i],1)) * pop[i]
         rhat[i,j] <- q[i,j] * n[i,j]
         dv[i,j] <- 2 * (r[i,j]*(log(r[i,j])-log(rhat[i,j])) + (n[i,j]-r[i,j])*(log(n[i,j]-r[i,j]) - log(n[i,j]-rhat[i,j])))
         }
   dev[i] <- sum(dv[i,1:nc[i]-1])

   delta[i] ~ dnorm(md[i], prec)
   md[i] <- d[t[i]] - d[b[i]] + equals(m[i],3) * sw[i]

   for (j in 2:nc[i]) {
      p[i,C[i,j]] <- 1 - phi.adj[i,j]
      phi.adj[i,j] <- phi(theta[i,j-1])
      }
   }

   for(k in 2:N) {
      sw[k]<- (delta[k-1] - d[t[k-1]] + d[b[k-1]]) / 2
      }
   totresdev <- sum(dev[])
   z[1] <- 0
   for (j in 2:Cmax-1) {
      z.aux[j] ~ dunif(0,5)
      z[j] <- z[j-1] + z.aux[j]
      }

   for(i in 1:ns){ mu[i] ~ dnorm(0,0.0001) }

   d[1] <- 0
   beta[1] <- 0
   for (k in 2:nt){
      d[k] ~ dnorm(0,0.00001)
```

```
        beta[k] <- B
        }

    betaplac ~ dnorm(0,0.00001)
    tau~dunif(0,2)
    tau.sq<-tau*tau
    prec<-1/(tau.sq)

#baseline mu - based on average of the 31 trials including it.
    for (i in 1:31) { mu1[i]<-mu[i]*equals(b[i*2-1],1) }
    for (i in 1:6) { mu1[31+i]<-mu[31+i]*equals(b[60+i*3],1) }

    A<-sum(mu1[])/31
    B ~ dnorm(0,0.0001)

# calculate prob of achieving PASI50/75/90 on treat k for adults (Ta) and children (Tc)
    for (k in 1:nt) {
        for (j in 1: Cmax-1) {
            Ta[j,k] <- 1 - phi(A + d[k] + z[j])
            Tc[j,k] <- 1 - phi(A + d[k] + z[j] + B)
            }
        }

# calculate RR PASI50,75,90 on treat k
    for (c in 1:(nt-1)) {
        for (k in (c+1):nt) {
            for (j in 1: Cmax-1) {
                RRa[j,c,k] <- Ta[j,k]/Ta[j,c]
                RRc[j,c,k] <- Tc[j,k]/Tc[j,c]
                }
            }
        }

}
```